
SPARQL
An RDF Query Language



SPARQL
l SPARQL is a recursive acronym for

SPARQL Protocol And Rdf Query Language

l SPARQL is the SQL for RDF
l Example query suitable for DBpedia

# find countries and their languages

PREFIX dbo: <http://dbpedia.org/ontology/>
SELECT * WHERE {

?country a dbo:Country;
dbo:officialLanguage ?lang .

} 
LIMIT 10



SPARQL History

l Several RDF query languages were developed 
prior to SPARQL

l W3C RDF Data Access Working Group (DAWG) 
worked out SPARQL 2005-2008

l Became a W3C recommendation in Jan 2008
l SPARQL 1.1 (2013) is the current standard 
l Support for many prog. languages available
l W3 SPARQL 1.2 Community Group established 

in 2019 to explore extensions

https://www.w3.org/TR/sparql11-query/
https://www.w3.org/community/sparql-12/


Typical Architecture

SPARQL endpoint receives queries and requests via 
HTTP from programs or GUIs, accesses associated 
RDF triple store and returns result, e.g., data

Some SPARQL endpoints

There are many public endpoints, e.g.
lDbpedia:  https://dbpedia.ort/sparql/
lWikidata: https://query.wikidata.org/sparql
lDBLP: https://dblp.l3s.de/d2r/sparql

lSee W3C’s list of currently alive SPARQL 
endpoints

It’s not hard to set up your own, e.g.
l UMBC cybersecurity knowledge graph: 

http://eb4.cs.umbc.edu:9090/ckg/query/

https://dblp.l3s.de/d2r/sparql
https://www.w3.org/wiki/SparqlEndpoints


Endpoint GUIs

lSome endpoints offer their own SPARQL GUI 
you can use to enter ad hoc queries 

lThey may use the same URL as the REST 
interface and rely on the protocol to know 
when it’s a person and when a query
– Dbpedia: http://dbpedia.org/sparql/
– Wikidata: https://query.wikidata.org/
– DBLP: https://dblp.l3s.de/d2r/snorql/

http://dbpedia.org/sparql/
https://query.wikidata.org/
https://dblp.l3s.de/d2r/snorql/


General SPARQL GUIs

lYou can also access or run a general SPARQL 
GUI that can talk to any SPARQL endpoint

lA nice example is YASGUI, which has a 
public resource: https://yqagui.org/ and is 
available to download

lAnother open-source GUI is Twinkle

https://yqagui.org/
https://doc.yasgui.org/
http://www.ldodds.com/projects/twinkle/


YASGUI: Yet Another SPARQL GUI

https://yasgui.org



SPARQL query structure
l Prefix declarations for 

abbreviating URIs
l Dataset definition: what 

RDF graph(s) are being 
queried

l Result clause: what 
information to return 
from the query

l Query pattern: what to 
query for in dataset

l Query modifiers, slicing, 
ordering, rearranging 
query results

# prefix declarations
PREFIX ex: <http://example.com/rdf/>  …

# optional named graph source
FROM ... 

# result clause (select,ask,update…)
SELECT ... 

# query pattern
WHERE { ... } 
# query modifiers
ORDER BY ...
GROUP BY ….
LIMIT 100 



Basic SPARQL Query Forms 

lSELECT 
Returns all, or a subset of, the variables bound in a 
query pattern match

lASK
Returns boolean indicating whether a query pattern 
matches or not

lDESCRIBE 
Returns an RDF graph describing resources found

lCONSTRUCT 
Returns an RDF graph constructed by substituting 
variable bindings in a set of triple templates



SPARQL protocol parameters

lTo use this query, we need to know]
– What endpoint (URL) to send it to
– How we want the results encoded (JSON, XML, …)
– … other parameters …

lThese are set in GUI or your program
– Except for the endpoint, all have defaults

lCan even query with the unix curl command:
curl http://dbpedia.org/sparql/ --data-urlencode query='PREFIX yago: 
<http://dbpedia.org/class/yago/> SELECT * WHERE {?city rdf:type
yago:WikicatCitiesInMaryland.}'



Exploring SPARQL with DBpedia

l DBpedia is a knowledge graph extracted from
different Wikipedia sites 

l Started in 2007, it continued to develop and offer 
services based on it

l Explore it in your browser in a human-readable form
l Query it using a public SPARQL endpoint to collect 

data
l Use services like Dbpedia Spotlight to get entities and 

concepts from text
l Download its data as JSON objects for your own use



Let’s find data about cities in MD

l We need to understand how DBpedia models data 
about cites

l We can view the ontology with its ~700 classes and
~2,800 properties

l And/or examine familiar entities, like Baltimore by
– Doing a web search on dbpedia Baltimore
– Clicking on links in the resulting page

http://mappings.dbpedia.org/server/ontology/classes/
http://dbpedia.org/page/Baltimore


Baltimore in Dbpedia (1)

final URL part is Wikipedia name

Property value pairs for this subject

DBO: is used as the prefix for the DBpedia ontology



Baltimore in Dbpedia (2)

Scroll down to find the rdf:type
Property to see Baltimore’s
types



Baltimore in Dbpedia (3)

This looks like the type we want!

Note: yago provides an ontology 
derived from Wikipedia with >  
10M entities.

For example, it induces types from 
Wikipedia category pages.

https://en.wikipedia.org/wiki/YAGO_(database)


A Query: Maryland Cities

# find URIs for cities in Maryland
PREFIX yago:  <http://dbpedia.org/class/yago/>
SELECT * WHERE {

?city a yago:WikicatCitiesInMaryland
}

http://dbpedia.org/class/yago/WikicatCitiesInMaryland


Maryland Cities and population

# get cities in MD and their populations
PREFIX yago: <http://dbpedia.org/class/yago/>t
PREFIX dbo: <http://dbpedia.org/ontology/>
SELECT * WHERE {

?city a yago:WikicatCitiesInMaryland;
dbo:populationTotal ?population .

}

http://dbpedia.org/class/yago/
http://dbpedia.org/class/yago/WikicatCitiesInMaryland


Maryland cities, population, names

# this returns names in multiple languages L
PREFIX yago:  <http://dbpedia.org/class/yago/>
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
SELECT ?city ?name ?population WHERE {

?city a yago:WikicatCitiesInMaryland;
dbo:populationTotal ?population ;
rdfs:label ?name .

}

http://dbpedia.org/class/yago/
http://dbpedia.org/ontology/
http://dbpedia.org/class/yago/WikicatCitiesInMaryland


Just the @en names, w/o lang tag
# FILTER gives conditions that must be true
# LANG(x) returns string’s language tag or ””
# STR(x) returns a string’s value, i.e. w/o language tag
PREFIX yago:  <http://dbpedia.org/class/yago/>
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
select (str(?name) as ?name) ?population where {
?city a yago:WikicatCitiesInMaryland;

dbo:populationTotal ?population;
rdfs:label ?name .

FILTER (LANG(?name) = "en")
}

http://dbpedia.org/class/yago/
http://dbpedia.org/ontology/


Order results by population (descending)

# sort results by population
PREFIX yago:  http://dbpedia.org/class/yago/
PREFIX dbo: <http://dbpedia.org/ontology/>

select str(?name) ?population where {
?city a yago:WikicatCitiesInMaryland;

dbo:populationTotal ?population;
rdfs:label ?name .

FILTER (LANG(?name) = "en")
}
ORDER BY DESC(?population)

http://dbpedia.org/class/yago/


Wait, where’s Catonsville? L

lMD’s government focused on counties
lCatonsville not considered a city – it has no 

government
lWe need another category of place

– Census designated place?  Populated Place?
lPopulated places include counties & regions; 

let’s use census designated place
lBut some ‘real’ cities in Maryland are not listed 

as census designated places and some are 

https://en.wikipedia.org/wiki/Census-designated_place
https://en.wikipedia.org/wiki/Human_settlement


UNION operator is OR
PREFIX yago:  <http://dbpedia.org/class/yago/>
PREFIX dbo: http://dbpedia.org/ontology/
PREFIX dbr: <http://dbpedia.org/resource/>

SELECT str(?name) ?population where {
{?city dbo:type dbr:Census-designated_place;    

dbo:isPartOf dbr:Maryland .}
UNION
{?city a yago:WikicatCitiesInMaryland . }
?city dbo:populationTotal ?population; rdfs:label ?name .
FILTER (LANG(?name) = "en")

}
ORDER BY DESC(?population)

http://dbpedia.org/class/yago/
http://dbpedia.org/ontology/


Now we have duplicate entries L

lThis happens because:
– Some “cities” are just in WikicatCitiesInMaryland
– Some are just in Census-designated_places
– Some are in both

lSPARQL’s procedure finds all ways to satisfy a query, 
and for each one, records the variable bindings

lWe add DISTINCT to  get SPARQL to remove 
duplicate bindings from the results



DISTINCT produces unique results

PREFIX yago:  <http://dbpedia.org/class/yago/>
PREFIX dbo: http://dbpedia.org/ontology/
PREFIX dbr: <http://dbpedia.org/resource/>
SELECT DISTINCT str(?name) ?population where {

{?city dbo:type dbr:Census-designated_place;    
dbo:isPartOf dbr:Maryland .}

UNION
{?city a yago:WikicatCitiesInMaryland . }
?city dbo:populationTotal ?population;

rdfs:label ?name .
FILTER (LANG(?name) = "en")

}
ORDER BY DESC(?population)

http://dbpedia.org/class/yago/
http://dbpedia.org/ontology/


Some cities are missing L

lExperimentation with query showed there are 
427 entities in MD that are either census 
designated places or cities

lOnly get 411 because nine have no population 
and one has neither a population nor a label
– Typical of a large and somewhat noisy knowledge 

graph created from crowdsourced data
lSPARQL’s OPIONAL directive to the rescue 



OPTIONAL handles missing data
PREFIX yago:  <http://dbpedia.org/class/yago/>
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX dbr: <http://dbpedia.org/resource/>
select DISTINCT str(?name) ?population where {

{?city dbo:type dbr:Census-designated_place;    
dbo:isPartOf dbr:Maryland .}

UNION
{?city a yago:WikicatCitiesInMaryland . }
OPTIONAL {?city dbo:populationTotal ?population.}
OPTIONAL {?city rdfs:label ?name . FILTER (LANG(?name) = 

"en") }
}
ORDER BY DESC(?population)

http://dbpedia.org/class/yago/
http://dbpedia.org/ontology/


Handling queries with many results

l Endpoints typically have limits on a query’s runtime or 
the number of results it can return

l You can use the LIMIT and OFFSET query modifiers to 
manage large queries

l Suppose we want to find all types that DBpedia uses

SELECT distinct ?type WHERE {?x a ?type.}

lDBpedia’s public endpoint limits queries 
to 10K results



Get the first 10K



Get the second 10K with OFFSET



A simple 
program 

gets 
them all

from SPARQLWrapper import SPARQLWrapper, JSON
default_endpoint = "http://dbpedia.org/sparql"
type_query = """SELECT DISTINCT ?class WHERE {{?x a ?class}} LIMIT {LIM} OFFSET {OFF}"""
def getall(query, endpoint=default_endpoint):

limit = 10000
offset = total = 0
found = limit
tuples = []
sparql = SPARQLWrapper(endpoint)
sparql.setReturnFormat('json')
while found == limit:  # keep going until we don't get limit results

q = query.format(LIM=limit, OFF=offset)
sparql.setQuery(q)
results = sparql.query().convert()
found = 0
for result in results["results"]["bindings"]:

found += 1
tuples.append(tuple([str(v['value']) for v in result.values()]))

print('Found', found, 'results')
total = total + found
offset = offset + limit

return tuples



ASK query

lAn ASK query returns True if it can be satisfied 
and False if not

lWas Barack Obama born in the US?
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX dbr: <http://dbpedia.org/resource/>
ask WHERE {
{dbr:Barack_Obama dbo:birthPlace dbr:United_States}
UNION
{dbr:Barack_Obama dbo:birthPlace ?x .
?x dbo:isPartOf*/dbo:country dbr:United_States }

} 



DESCRIBE Query

l “Describe ?x” means “tell me everything you 
know about ?x

lExample: Describe Alan Turing …
DESCRIBE <http://dbpedia.org/resource/Alan_Turing>
-- or –
PREFIX dbr: <http://dbpedia.org/resource/>
DESCRIBE dbr:Alan_Turing

l Returns a collection of ~1500 triples in which 
dbr:Alan_Turing is either the subject or object



Describes’s results?

lThe DAWG did not reach a consensus on what 
describe should return

lPossibilities include
– All triples where the variable bindings are 

mentioned
– All triples where the bindings are the subject
– Something else

lWhat is useful might depend on the application 
or the amount of data involved

lSo it was left to the implementation



DESCRIBE Query (2)

lDescribe the film “Double Indemnity”
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
PREFIX dbo: <http://dbpedia.org/ontology/>
describe ?x WHERE {

?x a dbo:Film; foaf:name ?filmName .
FILTER (STR(?filmName) = "Double Indemnity")

}

l Returns a collection of ~500 triples 



DESCRIBE Query (3)

lDescribe can return triples about
multiple entities

lDescribe films directed by Billy Wilder
PREFIX dbo: http://dbpedia.org/ontology/
PREFIX dbr: <http://dbpedia.org/resource/>
describe ?x WHERE {
?x a dbo:Film; dbo:director dbr:Billy_Wilder.

}

l Returns a collection of ~8400 triples about the 
27 films he directed

http://dbpedia.org/ontology/


DESCRIBE Query (4)
lDescribe can return triples about

multiple entities, but you can limit
the number

lDescribe films directed by Billy Wilder
PREFIX dbo: http://dbpedia.org/ontology/
PREFIX dbr: <http://dbpedia.org/resource/>
describe ?x WHERE {
?x a dbo:Film; dbo:director dbr:Billy_Wilder.

} LIMIT 1

l Returns a collection of ~500 triples about just 
one  film, The Apartment.

http://dbpedia.org/ontology/


Construct query (1)

lConstruct queries return graphs as results, e.g., 
film directors and the actors they’ve directed
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX ex: <http://example.org/>
CONSTRUCT {?director ex:directed ?actor}
WHERE {?film a dbo:Film; 

dbo:director ?director;
dbo:starring ?actor}

lReturns a graph with ~21,000 triples



On construct

lHaving a result form that produces an RDF 
graph is a good idea

l It enables on to construct systems by using the 
output of one SPARQL query as the data over 
which another query works

lThis kind of capability was a powerful one for 
relational databases



SPARQL 1.1 
allows using  
alternative 
properties 

separated by 
vertical bar

Construct query (2)

lActors and directors or producers
they’ve worked for
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX ex: <http://example.org/>
Construct {?actor ex:workedFor ?directorOrProducer}
WHERE {
?film a dbo:Film; 

dbo:director|dbo:producer ?directorOrProducer;
dbo:starring ?actor} 

lReturns a graph with ~31,000 triples



Example: finding missing inverses

l DBpedia is missing many inverse relations, including 
more than 10k missing spouse relations

l This creates a graph of all the missing ones, which can 
be added back to the KG via UPDATE ADD

PREFIX dbo: <http://dbpedia.org/ontology/>
CONSTRUCT { ?p2 dbo:spouse ?p1. }
WHERE {?p1 dbo:spouse ?p2.

FILTER NOT EXISTS {?p2 dbo:spouse ?p1}}

lNot the NOT EXISTS operator that succeeds iff
its graph pattern is not satisfiable



RDF Named graphs

l Having multiple RDF graphs in a single 
document/repository and naming them with URIs

l Provides useful additional functionality built on top of 
the RDF Recommendations

l SPARQL queries can involve several graphs, a 
background one and multiple named ones, e.g.:

SELECT ?who ?g ?mbox
FROM <http://example.org/dft.ttl>
FROM NAMED <http://example.org/alice>
FROM NAMED <http://example.org/bob>
WHERE
{ ?g dc:publisher ?who .

GRAPH ?g { ?x foaf:mbox ?mbox }
}



UPDATE QUERIES

l Simple insert
INSERT DATA { :book1 :title "A new book" ; :creator 
"A.N.Other" . }

l Simple delete
DELETE DATA { :book1 dc:title "A new book" . }

l Combine the two for a modification, optionally guided 
by the results of a graph pattern
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
DELETE { ?person foaf:givenName 'Bill’ } 
INSERT { ?person foaf:givenName 'William’ } 
WHERE { ?person foaf:givenName 'Bill' } 



Aggregation Operators

lSPARQL 1.1 added many aggregation 
operators, like count, min, max, …

lGenerally used in the results specification
PREFIX dbo: <http://dbpedia.org/ontology/>
SELECT (COUNT(?film) AS ?numberOfFilms)

WHERE {?film a dbo:Film .}

lThis finds 129,980 films



Group by

lGROUP BY breaks the query's result set into 
groups before applying the aggregate functions

lFind BO’s properties and group them by 
property and find the number in each group
PREFIX dbr: <http://dbpedia.org/resource/>
PREFIX dbo: <http://dbpedia.org/ontology/>
SELECT ?p (COUNT(?p) as ?number)
WHERE { dbr:Barack_Obama ?p ?o }
GROUP BY ?p ORDER BY DESC(count(?p))



Inference via SPARQL

This query adds inverse spouse relations that don’t 
already exist:

PREFIX dbo: <http://dbpedia.org/ontology/>
INSERT { ?p2 dbo:spouse ?p1. }
WHERE {?p1 dbo:spouse ?p2.

FILTER NOT EXISTS {?p2 dbo:spouse ?p1}}
l SPIN and SHACL are systems to represent simple 

constraint & inference rules that are done by sparql
l A big feature is that the rules are represented in the 

graph

http://spinrdf.org/
https://www.w3.org/TR/shacl/


lSPARQ 1.1 added many more
features …
– Subqueries
– Negation: MINUS
– Federated queries that access multiple endpoints

lData you want to extract from an RDF graph 
can probably be returned by one query
– Might be a complicated one, though …

lSearch web for SPARQL tricks or this book

SPARQL 1.1 Additions


